

# Creating Reports Again and Again ... and Again with Macro

|   |          |
|---|----------|
| <b>Creating Reports Demo.....</b>   | <b>3</b> |
| Demonstration: Creating Multiple Reports with %DO loops and Sending to Microsoft<br>Excel ..... | 4        |



# Creating Reports Demo

---

The purpose of this demo is to use the **RatingsReport** macro definition to create a variety of reports based on the **ProductReviews** table and send them to Microsoft Excel.


**Note:** This program assumes that users will spell state names correctly.

Here are the three report varieties:

- one summarized report that shows overall information about ratings
- a report for every state showing overall information about ratings
- one report for a selected state showing overall information about ratings

Here's how to access the files for the demo:

1. From the Windows taskbar, launch Google Chrome. When the browser opens, select **SAS Studio** from the bookmarks bar.
2. The user ID and password should be prefilled. If not perform these steps:
  - a. Enter **student** in the **User ID** field.
  - b. Enter **Metadata0** in the **Password** field.
3. Click **Sign in**.
4. When asked “Do you want to opt in to all of your assumable groups?” click **Yes**.
5. On the left, in the Explorer pane, select **Server > Home > SAS Explorer** to access the three files used for this demo.



**Note:** The programs currently have LIBNAME statement paths to match the path listed above. If you have a different path, you will need to adjust the LIBNAME statement accordingly.



## Creating Multiple Reports with %DO loops and Sending to Microsoft Excel

1. Open **Starter Macro Definition Program**. Run the program to create the **RatingsReport** macro definition and see the results of the macro definition call.

| Product Group     | Average Overall Rating |
|-------------------|------------------------|
| Eclipse Clothing  | 3.01343                |
| Eclipse Shoes     | 2.99893                |
| Green Tomato      | 2.99802                |
| Knitwear          | 2.98466                |
| LSF               | 2.97355                |
| Leisure           | 2.99501                |
| Massif            | 2.95109                |
| Orion             | 2.98590                |
| Orion Clothing    | 3.00188                |
| Osprey            | 3.02465                |
| Shoes             | 2.99971                |
| Shorts            | 2.92915                |
| Stockings & Socks | 2.95863                |
| Street Wear       | 3.01651                |
| T-Shirts          | 3.02077                |
| Tracker Clothes   | 2.99670                |
| Tracker Shoes     | 3.03441                |
| Twain             | 2.86131                |

Ratings below 3 stars are shaded red

| Frequency                    | Table of Product_Group by Overall |       |       |       |      |       |
|------------------------------|-----------------------------------|-------|-------|-------|------|-------|
| Product_Group(Product Group) | Overall                           |       |       |       |      | Total |
|                              | 1                                 | 2     | 3     | 4     | 5    |       |
| Eclipse Clothing             | 1306                              | 3851  | 6801  | 3782  | 1456 | 17196 |
| Eclipse Shoes                | 909                               | 2411  | 4527  | 2421  | 898  | 11166 |
| Green Tomato                 | 90                                | 212   | 406   | 212   | 89   | 1009  |
| Knitwear                     | 314                               | 813   | 1368  | 796   | 295  | 3586  |
| LSF                          | 158                               | 457   | 772   | 437   | 142  | 1966  |
| Leisure                      | 155                               | 396   | 705   | 401   | 148  | 1805  |
| Massif                       | 23                                | 29    | 78    | 42    | 12   | 184   |
| Orion                        | 287                               | 716   | 1342  | 735   | 254  | 3334  |
| Orion Clothing               | 165                               | 475   | 847   | 475   | 167  | 2129  |
| Osprey                       | 126                               | 339   | 633   | 338   | 146  | 1582  |
| Shoes                        | 275                               | 756   | 1434  | 751   | 277  | 3493  |
| Shorts                       | 39                                | 124   | 204   | 87    | 40   | 494   |
| Stockings & Socks            | 102                               | 259   | 440   | 254   | 81   | 1136  |
| Street Wear                  | 158                               | 500   | 857   | 478   | 187  | 2180  |
| T-Shirts                     | 117                               | 252   | 518   | 313   | 100  | 1300  |
| Tracker Clothes              | 1083                              | 3142  | 5593  | 2968  | 1147 | 13933 |
| Tracker Shoes                | 307                               | 952   | 1790  | 1018  | 350  | 4417  |
| Twain                        | 31                                | 65    | 108   | 51    | 19   | 274   |
| Total                        | 5645                              | 15749 | 28423 | 15559 | 5808 | 71184 |

2. Add **State** as a positional parameter to the **RatingsReport** macro definition.

```
%macro RatingsReport(State, RatingType=Overall, StarLow=1, StarHigh=5);
```

3. Revise the macro definition text to run existing code if no value is entered for **State** when **RatingsReport** is called.
  - a. Under the %MACRO statement, add a %IF/%THEN/%DO statement to submit the code only when **State** is null
  - b. Before the %MEND statement, add a %END statement.

```
%macro RatingsReport(State, RatingType=Overall, StarLow=1, StarHigh=5);

%if &State= %then %do;

  /*-----Report 1 - Average Satisfaction Rating (Choose Rating Type)-----*/
  proc sql;
    create table Average&RatingType as
    select Product_Group, avg( &RatingType) as Average "Average &Ratingtype Rating"
    from productreviews
    group by Product_Group;
  quit;

  proc format;
    value underperform 0 - < 3 = "salmon";
  run;

  title1 "Average &Ratingtype Satisfaction Rating By Product Group";
  footnote "Ratings below 3 stars are shaded red";
  proc print data=Average&Ratingtype noobs label;
    var Product_Group Average / style={background=underperform.};
  run;
  footnote;

  /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

  title1 "Satisfaction Ratings by Product Group";
  title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";

  proc freq data=productreviews;
    tables Product_Group * &RatingType /norow nocol nopercnt ;
    where &StarLow <= &RatingType <= &StarHigh;
  run;
%end;

%mend ratingsreport;

%ratingsreport()
```

4. Submit the code to ensure that results remain the same.

5. Revise the macro definition text to run a new block of code if a state name is entered in the parenthesis when **RatingsReport** is called.
  - a. After the %IF/%THEN-%DO group, add a %ELSE/%DO statement and a %END statement.
  - b. Copy and paste the previous report code between the %ELSE/%DO and %END statements.
  - c. In the PROC SQL step, add a WHERE clause to filter the **State\_Province** column based on the value entered in the parentheses when **RatingsReport** is called
  - d. In the TITLE1 statement of the PROC PRINT report, add **and State** to the end of the title.
  - e. Add an additional title to the PROC PRINT report that lists the state name that is entered.
  - f. In the TITLE1 statement of the PROC FREQ report, add **and State** to the end of the title.
  - g. Add an additional title to the PROC FREQ report that lists the state name that is entered.
  - h. In the WHERE statement in the PROC FREQ step, include an additional filter for **State\_Province** to equal the value entered for **State**.

```
%end;

%else %do;

  /*-----Report 1 - Average Satisfaction Rating (Choose Rating Type)-----*/
  proc sql;
    create table Average&RatingType as
    select Product_Group, avg( &RatingType) as Average "Average &Ratingtype Rating"
    from productreviews
    where State_Province = "&State"
    group by Product_Group;
  quit;

  proc format;
    value underperform 0 - < 3 = "salmon";
  run;

  title1 "Average &Ratingtype Satisfaction Rating By Product Group and State";
  title2 "&State";
  footnote "Ratings below 3 stars are shaded red";
  proc print data=Average&Ratingtype noobs label;
    var Product_Group Average / style={background=underperform.};
  run;
  footnote;

  /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

  title1 "Satisfaction Ratings by Product Group and State";
  title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";
  title3 "&State";

  proc freq data=productreviews;
    tables Product_Group * &RatingType /norow nocol nopercnt ;
    where &StarLow <= &RatingType <= &StarHigh and State_Province="&State";
  run;
%end;

%mend ratingsreport;
```



8. Under the %ELSE/ %IF/%THEN-%DO statement, add PROC SQL code to create macro variables to represent each state name.

```
%else %if %upcase(&State)= DETAILS %then %do;

  proc sql;
    select distinct State_Province
    into :StateName1-
    from productreview;
  quit;

  /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

  title1 "Satisfaction Ratings by Product Group";
  title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";

  proc freq data=productreviews;
    tables Product_Group * &RatingType /norow nocol nopercnt ;
    where &StarLow <= &RatingType <= &StarHigh;
  run;
%end;
```

9. Create a %DO loop to loop through the report 2 code for as many states as there are in the data.
- After the PROC SQL step, add a %DO loop statement to go from 1 to **&sqllobs**.
  - At the end of the report 2 code, add a %END statement

```
%else %if %upcase(&State)= DETAILS %then %do;

  proc sql;
    select distinct State_Province
    into :StateName1-
    from productreview;
  quit;

  %do i=1 %to &sqllobs;
    /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

    title1 "Satisfaction Ratings by Product Group";
    title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";

    proc freq data=productreviews;
      tables Product_Group * &RatingType /norow nocol nopercnt ;
      where &StarLow <= &RatingType <= &StarHigh;
    run;
  %end;
%end;
```



10. Use an indirect reference to insert the state name into the WHERE statement for the PROC FREQ step and the additional TITLE statement.
  - a. Revise the WHERE statement in the PROC FREQ step to include the following:  
**where State\_Province="&&StateName&i"**
  - b. Add an additional TITLE statement to include the indirect reference.
  - c. Revise first title to include **and State**.

```
%else %if %upcase(&State)= DETAILS %then %do;

  proc sql;
    select distinct State_Province
    into :StateName1-
    from productreviews;
  quit;

  %do i=1 %to &sqllobs;
    /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

    title1 "Satisfaction Ratings by Product Group and State";
    title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";
    title3 "&&StateName&i";

    proc freq data=productreviews;
      tables Product_Group * &RatingType /norow nocol nopercnt ;
      where &StarLow <= &RatingType <= &StarHigh and State_Province="&&StateName&i" ;
    run;
  %end;
%end;
```

11. Run the program and confirm that it was successful. You should see a satisfaction report listed for each state in the table

```
%ratingsreport(Details)
```

12. Modify the code to have the reports sent to Excel when the macro definition is called.

- a. After the ODS NOPROCTITLE statement, add an ODS EXCEL statement to create an Excel file for the reports to be sent to. (Your path might differ.)

```
options mcompilenote=all;
ods noproctitle;

ods excel file="/home/student/SAS Explorer/ReviewReport.xlsx";

%macro RatingsReport(State, RatingType=Overall, StarLow=1, StarHigh=5);
```

- b. Add the ODS EXCEL CLOSE statement before the %MEND statement to close the Excel destination

```
%end;

ods excel close;

%mend ratingsreport;
```

13. Add ODS EXCEL options to include embedded titles, footnotes, and names for the worksheets for each version of the report.

- a. In the %DO group for **&State=**, add ODS EXCEL OPTIONS statements about the TITLE statements for the PROC PRINT and PROC FREQ steps .

```
%if &State= %then %do;

/*-----Report 1 - Average Satisfaction Rating (Choose Rating Type)-----*/
proc sql;
    create table Average&RatingType as
    select Product_Group, avg( &RatingType) as Average "Average &Ratingtype Rating"
    from productreviews
    group by Product_Group;
quit;

proc format;
    value underperform 0 - < 3 = "salmon";
run;

ods excel options (embedded_titles="yes" embedded_footnotes="yes" Sheet_Name="&RatingType Satisfaction");

title1 "Average &Ratingtype Satisfaction Rating By Product Group";
footnote "Ratings below 3 stars are shaded red";
proc print data=Average&Ratingtype noobs label;
    var Product_Group Average / style={background=underperform.};
run;
footnote;

/*----- Report 2 - Breakdown of Satisfaction Ratings -----*/
ods excel options (embedded_titles="yes" embedded_footnotes="yes" Sheet_Name="Star Distribution");

title1 "Satisfaction Ratings by Product Group";
title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";

proc freq data=productreviews;
    tables Product_Group * &RatingType /norow nocol nopercnt ;
    where &StarLow <= &RatingType <= &StarHigh;
run;
%end;
```

- b. In the %DO group for **%upcase(&State)=DETAILS**, add an ODS EXCEL OPTIONS statement about the TITLE statements for the PROC FREQ step.

```
%else %if %upcase(&State)= DETAILS %then %do;

proc sql;
    select distinct State_Province
    into :StateName1-
    from productreviews;
quit;

%do i=1 %to &sqlobs;
    /*----- Report 2 - Breakdown of Satisfaction Ratings -----*/
    ods excel options (embedded_titles="yes" embedded_footnotes="yes" Sheet_Name="Star Dist. &&StateName&i");

    title1 "Satisfaction Ratings by Product Group and State";
    title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";
    title3 "&&StateName&i";

    proc freq data=productreviews;
        tables Product_Group * &RatingType /norow nocol nopercnt ;
        where &StarLow <= &RatingType <= &StarHigh and State_Province="&&StateName&i" ;
    run;
%end;
%end;
```

- c. In the %ELSE/%DO group, add ODS EXCEL OPTIONS statements about the TITLE statements for the PROC PRINT and PROC FREQ steps.

```
%else %do;

/*-----Report 1 - Average Satisfaction Rating (Choose Rating Type)-----*/
proc sql;
  create table Average&RatingType as
  select Product_Group, avg( &RatingType) as Average "Average &Ratingtype Rating"
  from productreviews
  where State_Province = "&State"
  group by Product_Group;
quit;

proc format;
  value underperform 0 - < 3 = "salmon";
run;

ods excel options (embedded_titles="yes" embedded_footnotes="yes" Sheet_Name="&RatingType Satisfaction: &State");

title1 "Average &Ratingtype Satisfaction Rating By Product Group and State";
title2 "&State";
footnote "Ratings below 3 stars are shaded red";
proc print data=Average&Ratingtype noobs label;
  var Product_Group Average / style={background=underperform.};
run;
footnote;

/*----- Report 2 - Breakdown of Satisfaction Ratings -----*/

ods excel options (embedded_titles="yes" embedded_footnotes="yes" Sheet_Name="Star Dist: &State");

title1 "Satisfaction Ratings by Product Group and State";
title2 "&RatingType Satisfaction - &StarLow to &StarHigh Stars Ratings";
title3 "&State";

proc freq data=productreviews;
  tables Product_Group * &RatingType /norow nocol nopercnt ;
  where &StarLow <= &RatingType <= &StarHigh and State_Province="&State";
run;
%end;
```

14. Run the program calling **RatingsReport** with no value, a single state name, and then **Details**.

```
%ratingsreport(Details)
```

|                    | A   | B                 | C       | D                  | E   | F                   | G     | H                     | I | J |
|--------------------|---|-------------------|---------|--------------------|-----|---------------------|-------|-----------------------|---|---|
| 1                  | Satisfaction Ratings by Product Group and State |                   |         |                    |     |                     |       |                       |   |   |
| 2                  | Overall Satisfaction - 1 to 5 Stars Ratings     |                   |         |                    |     |                     |       |                       |   |   |
| 3                  | Alabama   |                   |         |                    |     |                     |       |                       |   |   |
| 4                  |   |                   |         |                    |     |                     |       |                       |   |   |
| 5                  | Table of Product_Group by Overall               |                   |         |                    |     |                     |       |                       |   |   |
| 6                  | Product_Group(Product Group)                    |                   | Overall |                    |     |                     |       |                       |   |   |
| 7                  | Frequency                                       | 1                 | 2       | 3                  | 4   | 5                   | Total |                       |   |   |
| 8                  | Eclipse Clothing                                | 18                | 42      | 67                 | 41  | 8                   | 176   |                       |   |   |
| 9                  | Eclipse Shoes                                   | 10                | 27      | 54                 | 22  | 12                  | 125   |                       |   |   |
| 10                 | Green Tomato                                    | 0                 | 1       | 8                  | 3   | 0                   | 12    |                       |   |   |
| 11                 | Knitwear  | 6                 | 9       | 14                 | 9   | 2                   | 40    |                       |   |   |
| 12                 | LSF   | 3                 | 3       | 10                 | 3   | 1                   | 20    |                       |   |   |
| 13                 | Leisure   | 3                 | 7       | 5                  | 6   | 0                   | 21    |                       |   |   |
| 14                 | Massif  | 0                 | 0       | 0                  | 2   | 0                   | 2     |                       |   |   |
| 15                 | Orion   | 4                 | 4       | 18                 | 6   | 6                   | 38    |                       |   |   |
| 16                 | Orion Clothing                                  | 4                 | 6       | 5                  | 4   | 0                   | 19    |                       |   |   |
| 17                 | Osprey  | 1                 | 6       | 2                  | 6   | 1                   | 16    |                       |   |   |
| 18                 | Shoes   | 2                 | 10      | 10                 | 7   | 3                   | 32    |                       |   |   |
| 19                 | Shorts  | 0                 | 0       | 3                  | 0   | 1                   | 4     |                       |   |   |
| 20                 | Stockings & Socks                               | 0                 | 2       | 5                  | 3   | 0                   | 10    |                       |   |   |
| 21                 | Street Wear                                     | 0                 | 4       | 6                  | 3   | 1                   | 14    |                       |   |   |
| 22                 | T-Shirts  | 1                 | 0       | 4                  | 5   | 0                   | 10    |                       |   |   |
| 23                 | Tracker Clothes                                 | 10                | 29      | 38                 | 27  | 12                  | 116   |                       |   |   |
| 24                 | Tracker Shoes                                   | 1                 | 5       | 23                 | 13  | 2                   | 44    |                       |   |   |
| 25                 | Twain   | 0                 | 0       | 2                  | 0   | 0                   | 2     |                       |   |   |
| 26                 | Total   | 63                | 155     | 274                | 160 | 49                  | 701   |                       |   |   |
| 27                 |   |                   |         |                    |     |                     |       |                       |   |   |
| Star Dist. Alabama |   | Star Dist. Alaska |         | Star Dist. Arizona |     | Star Dist. Arkansas |       | Star Dist. California |   |   |

End of Demonstration